# Többpontos numerikus deriváltak származtatása

Bartha Ferenc, SZTE, 2002. február

```
> restart;
```

**n-pontos formula:**`n:=5:`

próbáljuk ki más **n** értékekkel is....

```
> l := {seq(i,i=ceil(-(n-1)/2)..ceil((n-1)/2))}:
  printf("%a points= %a ",nops(l),l);
5 points= {-2, -1, 0, 1, 2}
```

**A pontok számához alkalmazható Taylor polinom:**

```
> L:=l minus {0}:m:=nops(L):
> fun:=taylor(f(x),x=0,m+1):
> fun:=subs(f(0)=f[0],fun);
> var:={seq((D@@i)(f)(0),i=1..m)}:
```

$$fun := f_0 + D(f)(0)\,x + \frac{1}{2}(D^{(2)})(f)(0)\,x^2 + \frac{1}{6}(D^{(3)})(f)(0)\,x^3 + \frac{1}{24}(D^{(4)})(f)(0)\,x^4 + O(x^5)$$

**Az egyenletrendszer elkészitése:**

```
> eq:={}:
  for i from 1 to m do
  ep:={subs(x=L[i]*h,f[L[i]]=convert(fun,polynom))}:print(ep[1]):
  eq:=eq union ep:
  od:
```

$$f_{-2} = f_0 - 2\,D(f)(0)\,h + 2\,(D^{(2)})(f)(0)\,h^2 - \frac{4}{3}(D^{(3)})(f)(0)\,h^3 + \frac{2}{3}(D^{(4)})(f)(0)\,h^4$$

$$f_{-1} = f_0 - D(f)(0)\,h + \frac{1}{2}(D^{(2)})(f)(0)\,h^2 - \frac{1}{6}(D^{(3)})(f)(0)\,h^3 + \frac{1}{24}(D^{(4)})(f)(0)\,h^4$$

$$f_1 = f_0 + D(f)(0)\,h + \frac{1}{2}(D^{(2)})(f)(0)\,h^2 + \frac{1}{6}(D^{(3)})(f)(0)\,h^3 + \frac{1}{24}(D^{(4)})(f)(0)\,h^4$$

$$f_2 = f_0 + 2\,D(f)(0)\,h + 2\,(D^{(2)})(f)(0)\,h^2 + \frac{4}{3}(D^{(3)})(f)(0)\,h^3 + \frac{2}{3}(D^{(4)})(f)(0)\,h^4$$

**és megoldása:**

```
> sol:=solve(eq,var):
  for i from 1 to m do
  print(sol[i]):
  od:
```

$$D(f)(0) = \frac{1}{12}\,\frac{-f_2 + 8\,f_1 + f_{-2} - 8\,f_{-1}}{h}$$

$$(D^{(2)})(f)(0) = -\frac{1}{12}\,\frac{-16\,f_1 + 30\,f_0 + f_2 + f_{-2} - 16\,f_{-1}}{h^2}$$

$$(D^{(3)})(f)(0) = -\frac{1}{2}\,\frac{-2\,f_{-1} + 2\,f_1 - f_2 + f_{-2}}{h^3}$$

$$(\mathrm{D}^{(4)})(f)(0) = \frac{f_{-2} + 6\,f_0 - 4\,f_1 + f_2 - 4\,f_{-1}}{h^4}$$

>

>